
1 Overview

This document lists all the changes needed to correct problems in the TTCN implementation of test case 9.4.5.5 which is part of the UTRAN test suite. Only essential changes to the TTCN are applied and documented in section 4.

With these changes applied the test case can be demonstrated to run with one or more 3G UEs 
(see section 6). Execution log files are provided as evidence. 

3 Verification Test Summary

Test Case:
TC_9_4_5_5
Test Group:
/9/UTRAN34_LocationUpdatingNIMTC.ttcn
ATS Version:
iwd-TTCN3-B2013-03_D15wk25 + essential modifications.

System Simulator used:
Rohde & Schwarz CMW500

UE used:
Intel 7260
Verification Status:
PASS
4 Corrections required for test case 9.4.5.5
4.1 Introduction

This section describes the changes required to make test case 9.4.5.5 run correctly with a 3G UE. All modifications are marked with label “WA#” in the TTCN comments column of the enclosed ATS [1].

The ATS version used as basis was UTRAN testsuite which is part of iwd-TTCN3-B2013-03_D15wk25 release. This testsuite provided by MCC160 contains UTRAN test cases.
4.2 f_TC_9_4_5_5_UTRAN()
	Record Name
	f_TC_9_4_5_5_UTRAN()

	Reason for change
	In the curent TTCN implementation, the function f_UTRAN34_Preamble is used to achieve the initial condition. This function expects the establishment cause set to registration for the rrc connection request. 

This is incorrect.

As per 24.008 Annex L, table L.1.1, the RRC Establishment cause for Any CS NAS procedure where the initiating uplink signalling message has the Device properties IE with low priority indicator set to "MS is configured for NAS signalling low priority" will be set as delayTolerantAccess
This needs to be changed.

	Summary of change
	Since the function f_UTRAN34_Preamble is used in many other places (for testcase other than NIMTC), created a new function f_UTRAN34_Preamble_NIMTC and called it instead.

	Source of change
	UTRAN34_GMM_NIMTC.ttcn

	
	


Before:
	function f_TC_9_4_5_5_UTRAN() runs on UTRAN_PTC
  { /* @sic Location updating / periodic per-device timer */
    //Initialise all UTRAN cells according to 34.108
    f_UTRAN34_Init();
    // Set CellA specific parameters
    f_UTRAN_CellInfo_SetNMO(utran34_CellA, tsc_NMO_II);
    f_UTRAN34_CellInfo_SetT3212(utran34_CellA, tsc_T3212_5);
    // CellA started
    f_UTRAN34_SS_CreateCellDCH(utran34_CellA);
    //Send default system information
    f_UTRAN34_SendDefSysInfo(utran34_CellA);
    //Equip UE with a USIM containing default values except for those shown in clause 9.4.5.5.4
    f_UT_USIM_Insert(UT, "34.123-1 clause 9.4.5.5.4");
    // Move UE to "Idle updated" on cell A for configuration
    f_UTRAN34_Preamble(utran34_CellA);

    //the UE is deactivated.
<<SKIPPED SECTION>>   



After:

	function f_TC_9_4_5_5_UTRAN() runs on UTRAN_PTC
  { /* @sic Location updating / periodic per-device timer */
    //Initialise all UTRAN cells according to 34.108
    f_UTRAN34_Init();
    // Set CellA specific parameters
    f_UTRAN_CellInfo_SetNMO(utran34_CellA, tsc_NMO_II);
    f_UTRAN34_CellInfo_SetT3212(utran34_CellA, tsc_T3212_5);
    // CellA started
    f_UTRAN34_SS_CreateCellDCH(utran34_CellA);
    //Send default system information
    f_UTRAN34_SendDefSysInfo(utran34_CellA);
    //Equip UE with a USIM containing default values except for those shown in clause 9.4.5.5.4
    f_UT_USIM_Insert(UT, "34.123-1 clause 9.4.5.5.4");
    // Move UE to "Idle updated" on cell A for configuration
    //WA#9_4_5_5 f_UTRAN34_Preamble(utran34_CellA);
    f_UTRAN34_Preamble_NIMTC (utran34_CellA);//WA#9_4_5_5
    //the UE is deactivated.
    <<SKIPPED SECTION>>



4.3 F_UTRAN34_Preamble_NIMTC
	Record Name
	f_UTRAN34_Preamble_NIMTC

	Reason for change
	Please see change 4.2 above

	Summary of change
	New function

	Source of change
	UTRAN34_CommonFunctions.ttcn

	
	


Before:
	function f_UTRAN34_Preamble_NIMTC(UTRAN_CellId_Type p_CellId) runs on UTRAN_PTC
  { /* @sic R5s130347: local GuardTimer sic@ */
    timer t_GuardTimer := tsc_GuardTimePreamble;
    t_GuardTimer.start;
    //  Switch cell power on
    f_UTRAN_SetCellPower(p_CellId, tsc_AttenuationServingCell);
    //  Switch on UE
    f_UT_SwitchOnUE(UT, false);
    //  Establish RRC Connection, perform Authentication, Security Mode procedure,
    //  Attach procedure and release RRC Connection
    f_UTRAN34_IdleUpdated(p_CellId, omit, delayTolerantAccess);
    t_GuardTimer.stop;
  }


4.4 fl_TC_9_4_5_5_Body()
	Record Name
	fl_TC_9_4_5_5_Body()

	Reason for change
	The current TTCN expects the establishment cause in the rrc connection request at Step 2 to be “registration”. However, as per 24.008 Annex L, table L.1.2, the RRC Establishment cause for Any PS NAS procedure where the initiating uplink signalling message has the Device properties IE with low priority indicator set to "MS is configured for NAS signalling low priority" will be set as delayTolerantAccess. Thus, the establishment cause in the rrc connection request at Step 3 should be delayTolerantAccess, not registration. This needs to be changed in the current TTCN. This also affects Step 10.
In current TTCN, the implementation for Steps 5 to 7 of Prose is not correct. The MS T3212 timer value is not being sent in the LAU ACCEPT message, which is incorrect. This needs to be implemented.

	Summary of change
	Changed the establishment cause in the Step 2 and Step 10 to delayTolerantAccess.

Re-worked the TTCN implementation of Steps 5 to 7. The original implementation is marked in blue. Called the function f_UTRAN34_LocationUpdate_WithoutLAUReq(utran34_CellA, v_RRC_DataInd_CS.ttcn_start, true, TMSI,omit, omit, true); and passed true as argument for GPRS timer. This function is parameterised for this reason (described in a separate change).

Also assigned intermediate verdicts to improve the test flow traceability

Please see screenshot below.

	Source of change
	UTRAN34_LocationUpdating_NIMTC.ttcn

	
	


Before:
	<<SKIPPED SECTION>>
//@siclog "Step 2-4" siclog@
    //"Establishment cause": Registration
    f_UTRAN34_RRC_ConnEst(utran34_CellA, registration);
    f_UTRAN_PreliminaryPass(__FILE__, __LINE__, "Test Case 9.4.5.5, Step 2");
    //If PS mode: Any subsequent GMM attach from the UE shall be accepted by the SS.
    if (pc_PS == true) {
      t_WaitForAttach.start;
      interleave { //Receive in any order Location Area Update or Attach Request
        //GMM attach
        [] U_Dc.receive(car_PS_InitDirectTransfer(utran_CellDedicated, tsc_RB3, cr_U_AttachReq(cr_AttachType(?, (tsc_I_AttachTypeGPRSAtt,
                                                                                                                 tsc_I_AttachTypeCombined)),
                                                                                               cr_MobileIdAny(omit),
                                                                                               cr_RAI(omit),
                                                                                               ?))) -> value v_RRC_DataInd_PS
          {
            v_GMM_Attach_received := true;
          }
        //Receive Location Area Update Request message
        [] U_Dc.receive(car_InitDirectTransfer(utran_CellDedicated, tsc_RB3, v_U_LocationUpdatingRequest_Step5)) -> value v_RRC_DataInd_CS
          {
            v_LAU_received := true;
         }
      }
    }
    else { // only CS

      //@siclog "Step 5 -7" siclog@

      //LOCATION UPDATING REQUEST

      //"location updating type": IMSI attach

      //"MS network feature support": 1 (MS supports the extended periodic timer in this domain)

      f_UTRAN34_LocationUpdate(utran_CellDedicated, true, TMSI, v_U_LocationUpdatingRequest_Step5);

    }

    if (v_GMM_Attach_received == true) {

      //SECURITY MODE COMMAND and SECURITY MODE COMPLETE

      f_UTRAN_GMM_Authentication();

      f_UTRAN_RRC_Security(utran34_CellA, true, ps_domain);

      f_UTRAN_AttachAccept(utran_CellDedicated, v_RRC_DataInd_PS.msg.attachRequest.attachType.attachType);// ATTACH ACCEPT, ATTACH COMPLETE

      t_WaitForAttach.stop;

      v_GMM_Attach_received := false;

    }

    if (v_LAU_received == true) {

      f_UTRAN34_LocationUpdate_WithoutLAUReq(utran_CellDedicated, v_RRC_DataInd_CS.ttcn_start, true, TMSI);

    }

    //@siclog "Step 8-9" siclog@
    //SS waits for the disconnection of the main signalling link.
    f_UTRAN34_RRC_ConnRel(utran34_CellA, cell_Dch);
    //@siclog "Step 10" siclog@
    //"Establishment cause": Registration.
    //This message shall be sent by the UE between 5 minutes 45 s and 6 minutes 15 s after step 9
    f_UTRAN34_RRC_RandAccFail(v_T3212_min);
    t_Wait.start;
    alt {
      [] U_TM.receive(car_RRC_ConnReq(utran34_CellA,
                                      tsc_RB0,
                                      cr_108_RRC_ConnReq_r8(registration))) -> value v_RLC_TR_DATA_IND 
        {
          v_InitialUE_Id := v_RLC_TR_DATA_IND.data.tM_message.uL_CCCH_Message.message_.rrcConnectionRequest.initialUE_Identity;
          f_UTRAN_CheckReleaseIndicator(v_RLC_TR_DATA_IND.data.tM_message.uL_CCCH_Message.message_.rrcConnectionRequest.v3d0NonCriticalExtensions.v4b0NonCriticalExtensions.rrcConnectionRequest_v4b0ext.accessStratumReleaseIndicator);
        }
      [] t_Wait.timeout {
        f_UTRAN_SetVerdictFailOrInconc(__FILE__, __LINE__, "RRC Connection Request not received within 5min45sec and 6min15sec after step 9");
      }
    }
    //@siclog "Step 11-12" siclog@
    f_UTRAN34_RRC_ConnEst_WithoutRRCConnectionReq(utran34_CellA, v_InitialUE_Id);
    //@siclog "Step 13-14" siclog@
    //LOCATION UPDATING REQUEST
    f_UTRAN34_LocationUpdate(utran34_CellA, true, TMSI, v_U_LocationUpdatingRequest_Step13);
    //@siclog "Step 15-16" siclog@
    //SS waits for the disconnection of the main signalling link.
    f_UTRAN34_RRC_ConnRel(utran34_CellA, cell_Dch);
  } // fl_TC_9_4_5_5_Body


After:

	<<SKIPPED SECTION>>
//@siclog "Step 2-4" siclog@
    //"Establishment cause": Registration
    f_UTRAN34_RRC_ConnEst(utran34_CellA, delayTolerantAccess/*WA#9_4_5_5 registration*/);
    f_UTRAN_PreliminaryPass(__FILE__, __LINE__, "Test Case 9.4.5.5, Step 2");
    //If PS mode: Any subsequent GMM attach from the UE shall be accepted by the SS.
    if (pc_PS == true) {
      t_WaitForAttach.start;
      interleave { //Receive in any order Location Area Update or Attach Request
        //GMM attach
        [] U_Dc.receive(car_PS_InitDirectTransfer(utran_CellDedicated, tsc_RB3, cr_U_AttachReq(cr_AttachType(?, (tsc_I_AttachTypeGPRSAtt,
                                                                                                                 tsc_I_AttachTypeCombined)),
                                                                                               cr_MobileIdAny(omit),
                                                                                               cr_RAI(omit),
                                                                                               ?))) -> value v_RRC_DataInd_PS
          {
            v_GMM_Attach_received := true;
          }
        //Receive Location Area Update Request message
        [] U_Dc.receive(car_InitDirectTransfer(utran_CellDedicated, tsc_RB3, v_U_LocationUpdatingRequest_Step5)) -> value v_RRC_DataInd_CS
          {
            v_LAU_received := true;
            f_UTRAN_PreliminaryPass(__FILE__, __LINE__, "Test Case 9.4.5.5, Step 5");
          }
      }
    }
    /*WA#9_4_5_5 : Change start
    else { // only CS
      //@siclog "Step 5 -7" siclog@
      //LOCATION UPDATING REQUEST
      //"location updating type": IMSI attach
      //"MS network feature support": 1 (MS supports the extended periodic timer in this domain)
      f_UTRAN34_LocationUpdate(utran_CellDedicated, true, TMSI, v_U_LocationUpdatingRequest_Step5);
    }
    if (v_GMM_Attach_received == true) {
      //SECURITY MODE COMMAND and SECURITY MODE COMPLETE
      f_UTRAN_GMM_Authentication();
      f_UTRAN_RRC_Security(utran34_CellA, true, ps_domain);
      f_UTRAN_AttachAccept(utran_CellDedicated, v_RRC_DataInd_PS.msg.attachRequest.attachType.attachType);// ATTACH ACCEPT, ATTACH COMPLETE
      t_WaitForAttach.stop;
      v_GMM_Attach_received := false;
    }
    if (v_LAU_received == true) {
      f_UTRAN34_LocationUpdate_WithoutLAUReq(utran_CellDedicated, v_RRC_DataInd_CS.ttcn_start, true, TMSI);
    }
    WA#9_4_5_5 : Change stop*/
    //WA#9_4_5_5 : Change start
    if (v_LAU_received == true) {
      f_UTRAN34_LocationUpdate_WithoutLAUReq(utran34_CellA, v_RRC_DataInd_CS.ttcn_start, true, TMSI,omit, omit, true);      
    }
    if (v_GMM_Attach_received == true) {
      //SECURITY MODE COMMAND and SECURITY MODE COMPLETE
      f_UTRAN_GMM_Authentication();
      f_UTRAN_RRC_Security(utran34_CellA, true, ps_domain);      
      f_UTRAN_AttachAccept(utran34_CellA, v_RRC_DataInd_PS.msg.attachRequest.attachType.attachType);// ATTACH ACCEPT, ATTACH COMPLETE
      t_WaitForAttach.stop;
      v_GMM_Attach_received := false;
    }
    //WA#9_4_5_5 : Change stop
    f_UTRAN_PreliminaryPass(__FILE__, __LINE__, "Test Case 9.4.5.5, Step 6"); //WA#9_4_5_5
    //@siclog "Step 8-9" siclog@
    //SS waits for the disconnection of the main signalling link.
    f_UTRAN34_RRC_ConnRel(utran34_CellA, cell_Dch);
    f_UTRAN_PreliminaryPass(__FILE__, __LINE__, "Test Case 9.4.5.5, Step 9"); //WA#9_4_5_5
    //@siclog "Step 10" siclog@
    //"Establishment cause": Registration.
    //This message shall be sent by the UE between 5 minutes 45 s and 6 minutes 15 s after step 9
    f_UTRAN34_RRC_RandAccFail(v_T3212_min);
    t_Wait.start;
    alt {
      [] U_TM.receive(car_RRC_ConnReq(utran34_CellA,
                                      tsc_RB0,
                                      cr_108_RRC_ConnReq_r8(delayTolerantAccess /*WA#9_4_5_5 registration*/))) -> value v_RLC_TR_DATA_IND 
        {
          v_InitialUE_Id := v_RLC_TR_DATA_IND.data.tM_message.uL_CCCH_Message.message_.rrcConnectionRequest.initialUE_Identity;
          f_UTRAN_CheckReleaseIndicator(v_RLC_TR_DATA_IND.data.tM_message.uL_CCCH_Message.message_.rrcConnectionRequest.v3d0NonCriticalExtensions.v4b0NonCriticalExtensions.rrcConnectionRequest_v4b0ext.accessStratumReleaseIndicator);
        }
      [] t_Wait.timeout {
        f_UTRAN_SetVerdictFailOrInconc(__FILE__, __LINE__, "RRC Connection Request not received within 5min45sec and 6min15sec after step 9");
      }
    }
    //@siclog "Step 11-12" siclog@
    f_UTRAN34_RRC_ConnEst_WithoutRRCConnectionReq(utran34_CellA, v_InitialUE_Id);
    f_UTRAN_PreliminaryPass(__FILE__, __LINE__, "Test Case 9.4.5.5, Step 12"); //WA#9_4_5_5
    //@siclog "Step 13-14" siclog@
    //LOCATION UPDATING REQUEST
    f_UTRAN34_LocationUpdate(utran34_CellA, true, TMSI, v_U_LocationUpdatingRequest_Step13);
    //@siclog "Step 15-16" siclog@
    //SS waits for the disconnection of the main signalling link.
    f_UTRAN34_RRC_ConnRel(utran34_CellA, cell_Dch);
  } // fl_TC_9_4_5_5_Body


4.5 TC_9_4_5_5()
	Record Name
	TC_9_4_5_5()

	Reason for change
	In the curent TTCN implementation, the guard timer for this testcse in insufficient and needs to be increased.

	Summary of change
	Increased the guard timer from 300s to 600s

	Source of change
	UTRAN_Testsuite.ttcn

	
	


Before:
	testcase TC_9_4_5_5() runs on MTC_UTRAN system SYSTEM_UTRAN {
    // @purpose
    //   Periodic routing area updating / accepted / per-device timer
    var UTRAN_PTC    v_UTRAN      := null;
    var GERAN_PTC    v_GERAN      := null;
    var EUTRA_PTC    v_EUTRA      := null;
    timer t_GuardTimer := int2float(400); 
    v_UTRAN := UTRAN_PTC.create alive;
    f_MTC_ConnectPTCs_UTRAN(v_UTRAN, v_GERAN, v_EUTRA);
    v_UTRAN.start(f_TC_9_4_5_5_UTRAN());
    t_GuardTimer.start;
    f_MTC_MainLoop(t_GuardTimer);
  }


After:
	testcase TC_9_4_5_5() runs on MTC_UTRAN system SYSTEM_UTRAN {
    // @purpose
    //   Periodic routing area updating / accepted / per-device timer
    var UTRAN_PTC    v_UTRAN      := null;
    var GERAN_PTC    v_GERAN      := null;
    var EUTRA_PTC    v_EUTRA      := null;
    timer t_GuardTimer := int2float(600); //WA#9_4_5_5 400->600
    v_UTRAN := UTRAN_PTC.create alive;
    f_MTC_ConnectPTCs_UTRAN(v_UTRAN, v_GERAN, v_EUTRA);
    v_UTRAN.start(f_TC_9_4_5_5_UTRAN());
    t_GuardTimer.start;
    f_MTC_MainLoop(t_GuardTimer);
  }


4.6 f_UTRAN34_LocationUpdate_WithoutLAUReq
	Record Name
	f_UTRAN34_LocationUpdate_WithoutLAUReq

	Reason for change
	In the curent TTCN implementation, there is no provision to send GPRS timer value (MS T3212 value) in the LAU ACCEPT message. This needs to be implemented.

	Summary of change
	Parameterised the function to check if the MS T£”!” timer needs to be passed. Basically, a variable p_GPRSTimer is now default set to false (This is in order not to disturb other usages where the timer value is not needed), and for testcase 9.4.5.5, when this function is called, passed value true for this variable.
Later in the function body, depending on the location update type and if this parameter was set to true, called templates which assign the MS T3212 timer. 

For this purpose, created new templates cs_U_LocUpdAcpTMSI_9_4_5_5, cs_LocUpdAcp_GPRS_Timer_Common and cs_GprsTimer3_9_4_5_5.  

	Source of change
	UTRAN34_CommonFunctions.ttcn

	
	


Before:
	function f_UTRAN34_LocationUpdate_WithoutLAUReq(UTRAN_CellId_Type p_CellId,
                                                  START_Value p_START_Value := '00000000000000000000'B,
                                                  boolean p_DoAuth := false,
                                                  LOCATION_UPDATE_ACCEPT_TYPE p_LocAcceptType := TMSI,
                                                  template (omit) PLMN_List p_EPLMN := omit, //@sic R5-134916 sic@
                                                  template (omit) RRCDataReqType p_U_MmInformation := omi)
    runs on UTRAN_PTC
  {
    var octetstring v_PLMN := f_U_Asn2Nas_PlmnId(f_UTRAN_CellInfo_GetPLMNIdentity(p_CellId));
    var octetstring v_LAC := f_UTRAN_CellInfo_GetLAC(p_CellId);
    var boolean v_NewKeyFlag := false;//No new key generation
    var UTRAN_SecurityInfo_Type v_SecurityInfo := f_UTRAN_Security_Get();
    var template (value) MobileIdentity v_MobileIdIMSI := f_Imsi2MobileIdentity(px_IMSI_Def);
    v_MobileIdIMSI.iei := '17'O; // Need to include iei in LUAccept, but function returns without iei
    // Store flag to be sent to LTE later;
    v_SecurityInfo.authKeys.CSinOtherRAT := true;
    f_UTRAN_Security_Set (v_SecurityInfo);
    f_UTRAN_SS_SecurityDownloadStart(cs_domain, p_START_Value);
    if (p_DoAuth) {
      f_UTRAN_MM_Authentication ();
      v_NewKeyFlag := true;
    }
    //Send SECURITY MODE COMMAND and receive SECURITY MODE COMPLETE
    f_UTRAN34_RRC_Security(p_CellId, v_NewKeyFlag, cs_domain);
    // Send MM Information to the UE during LAU procedure
    if (isvalue (p_U_MmInformation)) {
      U_Dc.send(cas_DataReq_CS(utran_CellDedicated,
                               tsc_RB3,
                               valueof(p_U_MmInformation)));
    }
    //Send Location Updating Accept message
    select (p_LocAcceptType) {
      case (TMSI) {
        
U_Dc.send (cas_DataReq_CS(utran_CellDedicated, tsc_RB3, cs_U_LocUpdAcpTMSI (v_PLMN, v_LAC, p_EPLMN)));
        
//Receive TMSI Reallocation Complete
        
U_Dc.receive(car_UplinkDirectTransfer(utran_CellDedicated, tsc_RB3, cr_U_TMSI_ReallocCmpl));
        
}     
      case (IMSI) {
        U_Dc.send (cas_DataReq_CS(utran_CellDedicated, tsc_RB3, cs_U_LocUpdAcpIMSI (v_PLMN, v_LAC, v_MobileIdIMSI, p_EPLMN)));
      }
      case (NONE) {
        U_Dc.send (cas_DataReq_CS(utran_CellDedicated, tsc_RB3, cs_U_LocUpdAcp_NoMobId (v_PLMN, v_LAC, p_EPLMN)));
      }
    }
  }


After:
	function f_UTRAN34_LocationUpdate_WithoutLAUReq(UTRAN_CellId_Type p_CellId,
                                                  START_Value p_START_Value := '00000000000000000000'B,
                                                  boolean p_DoAuth := false,
                                                  LOCATION_UPDATE_ACCEPT_TYPE p_LocAcceptType := TMSI,
                                                  template (omit) PLMN_List p_EPLMN := omit, //@sic R5-134916 sic@
                                                  template (omit) RRCDataReqType p_U_MmInformation := omit,
                                                  boolean p_GPRSTimer := false /*WA#9_4_5_5*/
                                                  )
    runs on UTRAN_PTC
  {
    var octetstring v_PLMN := f_U_Asn2Nas_PlmnId(f_UTRAN_CellInfo_GetPLMNIdentity(p_CellId));
    var octetstring v_LAC := f_UTRAN_CellInfo_GetLAC(p_CellId);
    var boolean v_NewKeyFlag := false;//No new key generation
    var UTRAN_SecurityInfo_Type v_SecurityInfo := f_UTRAN_Security_Get();
    var template (value) MobileIdentity v_MobileIdIMSI := f_Imsi2MobileIdentity(px_IMSI_Def);
    v_MobileIdIMSI.iei := '17'O; // Need to include iei in LUAccept, but function returns without iei
    // Store flag to be sent to LTE later;
    v_SecurityInfo.authKeys.CSinOtherRAT := true;
    f_UTRAN_Security_Set (v_SecurityInfo);
    f_UTRAN_SS_SecurityDownloadStart(cs_domain, p_START_Value);
    if (p_DoAuth) {
      f_UTRAN_MM_Authentication ();
      v_NewKeyFlag := true;
    }
    //Send SECURITY MODE COMMAND and receive SECURITY MODE COMPLETE
    f_UTRAN34_RRC_Security(p_CellId, v_NewKeyFlag, cs_domain);
    // Send MM Information to the UE during LAU procedure
    if (isvalue (p_U_MmInformation)) {
      U_Dc.send(cas_DataReq_CS(utran_CellDedicated,
                               tsc_RB3,
                               valueof(p_U_MmInformation)));
    }
    //Send Location Updating Accept message
    select (p_LocAcceptType) {
      case (TMSI) {
       if (p_GPRSTimer == false)
       {
        
U_Dc.send (cas_DataReq_CS(utran_CellDedicated, tsc_RB3, cs_U_LocUpdAcpTMSI (v_PLMN, v_LAC, p_EPLMN)));
        
//Receive TMSI Reallocation Complete
        
U_Dc.receive(car_UplinkDirectTransfer(utran_CellDedicated, tsc_RB3, cr_U_TMSI_ReallocCmpl));
        
}
       else
       
{
        
U_Dc.send (cas_DataReq_CS(utran_CellDedicated, tsc_RB3, cs_U_LocUpdAcpTMSI_9_4_5_5 (v_PLMN, v_LAC, p_EPLMN)));
        
//Receive TMSI Reallocation Complete
        
U_Dc.receive(car_UplinkDirectTransfer(utran_CellDedicated, tsc_RB3, cr_U_TMSI_ReallocCmpl));
       
}
      }
      case (IMSI) {
        U_Dc.send (cas_DataReq_CS(utran_CellDedicated, tsc_RB3, cs_U_LocUpdAcpIMSI (v_PLMN, v_LAC, v_MobileIdIMSI, p_EPLMN)));
      }
      case (NONE) {
        U_Dc.send (cas_DataReq_CS(utran_CellDedicated, tsc_RB3, cs_U_LocUpdAcp_NoMobId (v_PLMN, v_LAC, p_EPLMN)));
      }
    }
  }


4.7 cs_U_LocUpdAcpTMSI_9_4_5_5
	Record Name
	cs_U_LocUpdAcpTMSI_9_4_5_5

	Reason for change
	Please see change 4.6 above

	Summary of change
	New template

	Source of change
	UTRAN_NAS_PDU_Templates.ttcn

	
	


Before:
	template (value) RRCDataReqType cs_U_LocUpdAcpTMSI_9_4_5_5(octetstring p_PLMN,
                                                     octetstring p_LAC,
                                                     template (omit) PLMN_List p_EPLMN := omit) :=  //@sic R5-134916 sic@
  { /* @status    APPROVED (IMS, IMS_IRAT, LTE_A, LTE_IRAT, POS, SSNITZ, UTRAN) */
    locationUpdatingAccept := cs_LocUpdAcp_GPRS_Timer_Common (p_PLMN, p_LAC, cs_MobileIdTMSIorPTMSI ('17'O, px_TMSI_Def), p_EPLMN )
  };


4.8 cs_LocUpdAcp_GPRS_Timer_Common
	Record Name
	cs_LocUpdAcp_GPRS_Timer_Common

	Reason for change
	Please see change 4.6 above

	Summary of change
	New template

	Source of change
	NAS_24008Templates_IRAT.ttcn

	
	


Before:
	template (value) LOCATIONUPDATINGACCEPT cs_LocUpdAcp_GPRS_Timer_Common(octetstring p_PLMN,
                                                              octetstring p_LAC,
                                                              template (omit) MobileIdentity p_MobileId, // @sic R5-112706 sic@
                                                              template (omit) PLMN_List p_EPLMN
                                                               ) :=
  { /* @status    APPROVED (IMS, IMS_IRAT, LTE_A, LTE_IRAT, POS, SSNITZ, UTRAN) */
    skipIndicator   := '0000'B,
    mMProtocolDiscriminator := '0101'B,
    msgType         := '00000010'B,
    locAreaId       := cs_LAI(omit, p_PLMN, p_LAC),
    mobileId        := p_MobileId,
    followOnProceed := 'A1'O,
    cTSPerm         := omit,
    equivalentPLMN  := p_EPLMN,
    emergNumList    := omit,
    perMST3212      := cs_GprsTimer3_9_4_5_5 ( '101'B, '00110'B)   /* cl. 10.5.7.4a O TLV 3     IEI=0x35 @sic R5s120178 Baseline Moving sic@*/
  };


4.9 cs_GprsTimer3_9_4_5_5
	Record Name
	cs_GprsTimer3_9_4_5_5

	Reason for change
	Please see change 4.6 above

	Summary of change
	New template

	Source of change
	NAS_CommonTemplates.ttcn

	
	


Before:
	template (value) GPRS_Timer3 cs_GprsTimer3_9_4_5_5(GprsTimerUnit p_Unit,
                                             B5_Type p_Value) := cs_GPRS_Timer2_3_IEI('35'O, p_Unit, p_Value);  /* @status    APPROVED (IMS, IMS_IRAT, LTE, LTE_A, LTE_IRAT, POS) */



4.10 Branches executed in test case 9.4.5.5
The test case implementation was executed with Integrity and ciphering enabled.

5 Execution Log Files

5.1 Intel 7260 device
The Intel 7260 device passed this test case on Rohde & Schwarz CMW500. The documentation below is enclosed as evidence of the successful test case run [1]:

Test Case Execution log file: 
Rohde-Schwarz\tc_9_4_5_5_Intel_7260.log 
(Note: PICS/PIXIT settings are captured at the beginning of the TLI log)
In the log provided, masking is applied to prevent any disclosure of confidential information.
In the log file (in .txt / -html format) the complete test case execution can be seen. All TLI events are presented and message contents are fully decoded and can be verified. Preliminary verdicts and the final test case verdict can be seen in the log file
6 References
	[1]
	R5s150677
This archive comprises Execution log files, PICS/PIXIT files


